

## 1 Généralités

### 1.1 Liens historiques et pratiques

Créé au Xerox Parc, définitions : 1972, 1976, 1980

- Un point d'entrée général : [http://www.esug.org/wiki/pier/Smalltalk?\\_s=qI8P4-ZjisQSW0wA&\\_k=ExLkySmcSwfn&13](http://www.esug.org/wiki/pier/Smalltalk?_s=qI8P4-ZjisQSW0wA&_k=ExLkySmcSwfn&13).
- L'idée d'Alan Kay ([https://fr.wikipedia.org/wiki/Alan\\_Kay](https://fr.wikipedia.org/wiki/Alan_Kay)).
- <http://wiki.squeak.org/squeak/64> : “Smalltalk-80 : The Language and its Implementation” (Smalltalk Blue Book), par Adele Goldbeg et David Robson est le livre historique qui définit le langage Smalltalk-80 et le coeur de son implantation. Le “purple book” (<http://wiki.squeak.org/squeak/692>) est le même sans le chapitre 4 sur l'implantation.
- Le “Smalltalk orange book” : The Smalltalk-80 programming environment, de Adèle Goldberg et David Robson définit l'environnement de programmation historique Smalltalk. Vous y trouverez les idées partiellement matérialisées aujourd'hui dans Eclipse avec en plus son debugger avec “hot code replacement”.
- <http://www.esug.org/> : ESUG - European Smalltalk User Group.
- <http://www.pharo-project.org/home>. Pharo is a clean, innovative, open-source Smalltalk-inspired environment.
- [https://www.dailymotion.com/video/x3qedvo\\_fun-mooc-programmation-objet-immersive-en-pharo\\_school](https://www.dailymotion.com/video/x3qedvo_fun-mooc-programmation-objet-immersive-en-pharo_school), Mooc Pharo.
- <http://stephane.ducasse.free.fr/FreeBooks.html> : catalogue de livres sur Smalltalk dont certains en ligne.
- [www.squeak.org](http://www.squeak.org) : Squeak est un Smalltalk gratuit réalisé par une équipe dirigée par Alan Kay, l'inventeur premier du langage.
- <http://www.cincomsmalltalk.com/main/> : Cincom, Smalltalk industriel (ObjectStudio - Visualworks)
- <http://www.threeriversinstitute.org> : site de Kent Beck, un des pionniers du “Agile development”, méthodologie initialement développée autour de Smalltalk.
- <http://www.chimu.com/publications/JavaSmalltalkSyntax.html> : Une comparaison des syntaxes Java et Smalltalk.

- <http://www.csci.csusb.edu/dick/samples/smalltalk.syntax.html> : Une présentation concise de la syntaxe de Smalltalk.

## 1.2 Caractéristiques

Smalltalk est

- un **langage à objets** basé sur cinq idées,
- un environnement de programmation interactive complet : éditeur, compilateur, metteur au point, environnement de m
- C'est un environnement intégré de prototypage pour le développement rapide, agile, extrême, ...
- influencé par Simula (abstraction de données, surcharge des noms d'opérations, polymorphisme d'inclusion)
- influencé par Lisp
  - typage dynamique,
  - références,
  - gestion automatique de la mémoire, allocation dynamique et ramasse-miette,
  - style applicatif, toute instruction est une expression,
  - compilation en instruction d'une machine virtuelle

## Cinq idées

- Toute entité est un objet.

Un objet est une entité individuelle, repérée par une adresse unique, formé de plusieurs champs, connus par leurs noms (en nombre fixé) et contenant une valeur qui peut varier au cours du temps.

- tout objet appartient à (est instance) d'une classe qui définit sa structure et ses comportements.
- un comportement d'un objet est activé par un envoi de message.

Plus généralement, **tout calcul** en Smalltalk s'effectue par une succession d'envois de messages

- Toute classe est définie comme une spécialisation d'une autre. La relation de spécialisation définit un arbre d'héritage.

## 1.3 Interprétation, Compilation, machine virtuelle

Principe d'exécution : compilation des instructions en byteCodes ou instructions d'une machine virtuelle dédiée à la programmation par objets puis interprétation de ces instructions.

Machine virtuelle Smalltalk : ensemble d'instructions dédiées à la programmation par objets et interpréteur associé.

Le langage Java a repris ce schéma d'exécution.

## Exemple de méthode et de bytecode généré

```

1 fact
2   ^self = 0
3   ifTrue: [1]
4   ifFalse: [self * (self - 1) fact]

```

---

```

1 normal CompiledMethod numArgs=0 numTemps=0 frameSize=12
2 literals: (#fact )
3 1 <44> push self
4 2 <49> push 0
5 3 <A6> send =
6 4 <EC 07> jump true 13
7 6 <44> push self
8 7 <44> push self
9 8 <4A> push 1
10 9 <A1> send -
11 10 <70> send fact
12 11 <A8> send *
13 12 <66> pop
14 13 <60> push self; return

```

---

## 2 Syntaxe

### 2.1 constantes littérales

“Constantes” car leur valeur ne peut être modifiée et “littérales” car elles peuvent être entrées littéralement dans le texte des programmes.

- nombres : 3, 3.45, -3,
- caractères : \$a, \$M, \$\$
- chaînes : 'abc', 'the smalltalk system'
- symboles : #bill, #a22
- booléens true, false
- tableaux de constantes littérales : #(1 2 3 a22)
- commentaires : "ce commentaire sera ignoré par l'interpréteur"
- commentaires : #('vincent' 'francois' 'paul' 'autres')

### 2.2 Identificateurs

- Suite de chiffres et lettres commençant par une lettre.
- Manipulables dans le texte des programmes via leur symbole éponyme.

```
SmallInteger methodDictionary at: #fact
```

---

- Non typés (typage dynamique).

### 2.3 Affectation

```
i := 3
```

---

## 2.4 Séquence d'Instructions

Les instructions sont séparées par un point. Il y a une séquence implicite associée à tout bloc, corps de méthode par exemple.

```
i := 1. j := 2.
```

## 2.5 Retour à l'appelant

```
^33 "(return)"
```

Toute invocation de méthode rend une valeur. Le type void n'existe pas.

## 2.6 Envois de message

Définitions :

- Conceptuelle : demande à un objet d'exécuter un de ses comportements implanté sous la forme d'une méthode,
- Pratique : appel de méthode avec sélection selon le type dynamique du premier argument qualifié de receveur et syntaxiquement distingué.

Trois sortes de messages (distinction syntaxique) :

1. *unaires* (à un argument : le receveur)  
`1 class`  
`5 fact`
2. *binaires* (à deux arguments, syntaxe pratique pour les opérations arithmétiques en infixé)  
`1 + 2`
3. *"keywords"* à  $n, n \geq 2$  arguments dont le receveur.

Leur forme syntaxique (originale) permet de représenter un envoi de message comme une "petite conversation" (*small talk*) entre un objet et un autre.

```
1 log: 10
anArray at: 2 put: 3
monCalendrier enregistre: unEvenement le: unJour à: uneheure
```

Les familiers de Java traduiront en :

```
monCalendrier.enregistre(unEvenement, unJour, uneHeure)
```

Noms des méthodes utilisées dans ces exemples :

```
log:,
at:put:,
enregistre:le:à:.
```

- Précédence :  
unaire > binaire > keyword,  
exemple : `1 5 fact +` égale 121 et pas 720.

- Associativité A précedence égale les messages sont composés de gauche à droite.  
exemple :  
 $2 + 3 * 5 = 25$   
 $2 + (3 * 5) = 17$
- Cascade de messages  
`r s1; s2; s3.` est équivalent à `:r s1. r s2. r s3.`

## 2.7 Structures de contrôle

Il n'existe aucune forme syntaxique particulière pour les structures de contrôle ; le contrôle s'effectue par envoi de messages.

Le propre d'une structure de contrôle est d'avoir une politique non systématique d'évaluation de ses arguments. Implanter des structures de contrôle comme des méthodes standard nécessite donc que les arguments de ces méthodes ne soient pas évalués systématiquement au moment de l'appel. La solution Smalltalk à ce problème consiste utiliser des blocks, soit en position de receveur, soit en position d'argument.

Une fermeture lexicale est une fonction anonyme capturant son environnement lexical de définition (toute variable libre y est interprétée relativement à l'environnement dans lequel la fermeture a été définie).

La syntaxe de définition d'une fermeture est [ <parametre>\* | <instruction>\* ]

### 2.7.1 Exemples

#### Conditionnelle

```
(number \\ 2) = 0 ifTrue: [parity := 0] ifFalse: [parity := 1]
```

#### Boucle for

La boucle for utilise un block à un paramètre qui tient lieu de compteur de boucle

```
1 to: 20 do: [:i | Transcript show: i printString].
```

#### Itérateurs

```
untableau := #(3 5 7 9).  
untableau do: [:each | Transcript show: each; cr]
```

#### Formes de gestion d'exceptions

```
untableau findKey: x ifAbsent: [0]
```

## 2.7.2 Plus sur les fermetures

Les fermetures en Smalltalk sont implantées par la classe *BlockClosure*.

- Une fermeture est une constante littérale dont la valeur est elle-même.

```
[ :x | x + 1 ]  
= [ :x | x + 1 ]
```

- Une fermeture peut ainsi être utilisée comme valeur dans une affectation ou passée en argument, exemples :

```
V := [ :x | x + 1 ]  
(date = '24/12') ifTrue: [ Transcript show: 'Noël !!' ]
```

- Une fermeture est une fonction, que l'on peut donc invoquer, et comme c'est par ailleurs aussi un objet (tout est objet) on l'invoque en lui envoyant le message `value` :

```
[ 2 + 3 ] value  
= 5
```

- Si la fermeture possède des paramètres, on l'invoque en lui passant des arguments avec les méthodes `value:` , `value:value:`, etc (c'est moins beau qu'en *Scheme* mais c'est conforme au paradigme).

```
[ :i | i + 1 ] value: 33  
= 34
```

## 2.7.3 Fermeture et retour de méthode

Une fermeture capture le point de retour du block ou elle est créée.

```
1 test2Main  
2     self test2: [^1 + 2].  
3     Transcript show: '33333333'; cr.  
  
5 test2: aBlock  
6     Transcript show: '11111111'; cr.  
7     aBlock value.  
8     Transcript show: '22222222'; cr.  
  
10 Object test2Main "imprime 11111111 et rend 3"
```

A condition que le point de retour existe toujours lors de l'exécution du block.

```
1 test1Main  
2     | b |  
3     b := self test1.  
4     Transcript show: '11111111'; cr.  
5     Transcript show: b value.  
6     Transcript show: '22222222'; cr.
```

```

8 test1
9     ^[1 + 2] ""

```


- The receiver tried to return result to homeContext that no longer exists.

## 3 Pratique basique du langage

### 3.1 Classes et Méthodes

Toute classe est créée comme sous-classe d'une autre classe.

Variables d'instance, variables de classes (équivalent des "statiques" de C++ et Java), variables de *pool* (pas d'équivalent en Java).



**Figure (1)** – Définition de la classe Date

## Définition de méthodes d'instance




Figure (2) – Méthode d'instance < de la classe Date

## Définition de méthodes de classe


Méthodes invocables par envoi de message aux classes <sup>1</sup>

```
Date new. Date today.
```

Les méthodes de classe Smalltalk sont proposées en cohérence avec des méta-classes correspondantes. Toute classe C est instance d'une métaclasse, générée automatiquement au moment de la création de C et nommée "C class".

1. Les méthodes de classe sont en fait des méthodes d'instance définies sur les métaclasses, voir la section 4.3. En première approche, elles ressemblent aux *static* en C++ puis Java, mais elles sont en fait beaucoup plus puisque ce sont de vrais méthodes, redéfinissables comme les autres et invocables par envoi de message avec la sémantique standard.





**Figure (3)** – Méthode de classe “today” de la classe Date (En fait une méthode d’instance de la (méta)classe “Date class”)

## Différentes sortes de variables accessibles au sein des méthodes

Les variables accessibles au sein d’une méthode M :

- les paramètres de M,
- les variables temporaires de M, (exemple, réécrire *théta* en utilisant des variables temporaires)
- les variables d’instance de l’objet receveur O,
- les pseudo-variables *self*, *super*, *true*, *false*, *nil*.
- les variables de classe de la classe C de O (celle ou est définie M)
- les variables partagées entre la classe C et d’autres classes, dont les variables globales partagées entre toutes les classes,

## 3.2 Envoi de message

Procédé, réalisé à l’exécution, menant à l’exécution d’un méthode de même nom que le *selecteur*.

Consiste en la recherche de la méthode dans la classe du receveur du message puis en cas d’échec dans ses superclasses.

Diverses techniques d’optimisation de l’envoi de messages sont implantées dans les diverses machine virtuelle dont la principale est la techniques de cache.

## 3.3 Instantiation et initialisation des objets

- On instancie une classe en lui envoyant le message **new**.
- **new** est une méthode définie sur une superclasse commune à toutes les métaclasses (cf. métaclasses). **new** réalise l’allocation mémoire et rend la référence sur le nouvel objet.
- Exemple : `Date new, d := Date new.`

Il n’y a pas de constructeurs.

L'initialisation s'effectue via des méthodes standard, souvent invoquées dans des redéfinitions de la méthode de classe `new`.

Exemples : les méthodes `new` et `x:y:` de la classe `Point`.

### 3.4 Exemple

```
1 Object subclass: #Compteur
2   instanceVariableNames: 'valeur'
3   classVariableNames: ''
4   poolDictionaries: ''
5   category: 'Compteur-MVC'!

7 !Compteur methodsFor: 'acces lecture/ecriture'!
8 nombreUnites
9   ^valeur!

11 nombreUnites: n
12   valeur := n!!

14 !Compteur methodsFor: 'imprimer'!

16 printOn: aStream
17   aStream nextPut: $@.
18   self nombreUnites printOn: aStream!!
```

```
1 !Compteur methodsFor: 'operations'!

3 decr
4   self nombreUnites: valeur - 1!

6 incr
7   self nombreUnites: valeur + 1!

9 raz
10  self nombreUnites: 0!!
11  "-----"!

13 Compteur class
14   instanceVariableNames: ''!

16 !Compteur class methodsFor: 'creation'!
17 new
18   ^super new raz!!
```

### 3.5 Classe abstraite

Une classe ne peut être déclarée abstraite mais elle être rendue abstraite en redéfinissant la méthode `new` pour signaler une exception.

Ceci pose néanmoins de graves problèmes pour les futures sous-classes concrètes. (utilisation obligatoire de `basicNew`). Ceci est une limite de la solution Smalltalk pour les métaclasses.

### 3.6 Methode abstraite

Une méthode ne peut être déclarée abstraite mais elle être rendue abstraite en la définissant de la façon suivante :

```
method
  self subclassResponsibility
```

### 3.7 Sous-types et Héritage

- Héritage simple.
- Redéfinition sur la base du nom des méthodes. Toute définition sur une sous-classe d'une méthode de même nom existant sur une superclasse est une redéfinition.
- Typage dynamique :
  - pas de surcharge,
  - pas de problème de redéfinition covariante, pas de *downcast*,
  - aucun contrôle statique.
- redéfinition à corrélér à celle de tests systématiques et élaborés? (Plus globalement voir *Agile Development*).

### 3.8 Schémas de spécialisation

- Masquage : redéfinition d'une méthode sur une sous-classe.
- Masquage partiel : la pseudo-variable `super`.
- Paramétrage par spécialisation (pseudo-variable `self`) ou par composition, classiques.

### 3.9 Un exemple de hiérarchie : Les collections

```
1  ProtoObject #()
2    Object #()
3
4    Collection #()
5      Bag #('contents')
6      IdentityBag #()
7      CharacterSet #('map')
8      SequenceableCollection #()
9      ArrayedCollection #()
10     Array #()
11       ActionSequence #()
12       DependentsArray #()
13       WeakActionSequence #()
14       WeakArray #()
15       Array2D #('width' 'contents')
16       B3DPrimitiveVertexArray #()
17       Bitmap #()
18       Heap #('array' 'tally' 'sortBlock')
19       Interval #('start' 'stop' 'step')
20       TextLineInterval #('internalSpaces' 'paddingWidth' 'lineHeight' 'baseline')
21       LinkedList #('firstLink' 'lastLink')
22       Semaphore #('excessSignals')
23       MappedCollection #('domain' 'map')
24       OrderedCollection #('array' 'firstIndex' 'lastIndex')
25       GraphicSymbol #()
26       SortedCollection #('sortBlock')
27       UrlArgumentList #()
28       SourceFileArray #()
29       StandardSourceFileArray #('files')
30     Set #('tally' 'array')
31     Dictionary #()
```


```

32     HtmlAttributes #()
33     IdentityDictionary #()
34         SystemDictionary #('cachedClassNames')
35             Environment #('envtName' 'outerEnvt')
36                 SmalltalkEnvironment #()
37     LiteralDictionary #()
38     MethodDictionary #()
39     PluggableDictionary #('hashBlock' 'equalBlock')
40     WeakKeyDictionary #()
41         ExternalFormRegistry #('lockFlag')
42     WeakIdentityKeyDictionary #()
43     WeakValueDictionary #()
44     IdentitySet #()
45     PluggableSet #('hashBlock' 'equalBlock')
46     WeakSet #('flag')
47     SkipList #('sortBlock' 'pointers' 'numElements' 'level' 'splice')
48     IdentitySkipList #()
49     WeakRegistry #('valueDictionary' 'accessLock')

```

---

### 3.10 Héritage et description différentielle



**Figure (4)** – Hiérarchie d'héritage avec ajouts de fonctionnalités. Pour plus de détails sur la description différentielle, voir notes de cours “réutilisation et frameworks”.

## 4 La méta-programmation en Smalltalk

### 4.1 Définitions

**Réflexivité** : Capacité qu’a un système à donner à ses utilisateurs une représentation de lui-même en connexion causale avec sa représentation effective en machine.

**Entité de première classe** : entité ayant une représentation accessible dans un programme, que l’on peut référencer, manipuler et inspecter, et éventuellement modifier.

En Smalltalk, les classes, les méthodes compilées, les méthodes, certaines structures de contrôle, éventuellement la pile d'exécution, l'environnement de programmation, le compilateur ... sont des entités de première classe.

## 4.2 Méta-programmation basique

### 4.2.1 Les classes comme “rvalues” standards - Généricité

Le terme “généricité” désigne en Java la capacité à contraindre les structures de données de type collection à ne contenir que des éléments appartenant à des types identifiées dans le texte du programme.

En typage statique (voir les génériques *Java*, les templates *C++*) la vérification de telles contraintes peut être faite à la compilation.

Les classes comme “rvalues”, accompagnées d'une primitive de test de sous-typage, permettent le contrôle dynamique de types.

Exemple :

```
File subclass: #PileTypee
instanceVariableNames: 'typeElements'
classVariableNames: ''
category: 'TP1'

push: element
(element isKindOf: typeElements)
ifTrue: [ super push: element ]
ifFalse: [ self error: 'Impossible d''empiler ', element printString,
' dans une pile de ', typeElements printString]
```

### 4.2.2 Manipulation standard des objets primitifs (rock-bottom objects)

Il existe une classe représentant chacun des types primitifs d'objets : String, Float, SmallInteger, ...

Il est possible de modifier les méthodes de ces classes, attention ce peut être fort dangereux, ou d'en créer de nouvelles (exemple : définir la méthode **factorielle** sur la classe **Integer**).

Les objets primitifs sont utilisables comme les autres objets du système (différence avec Java), ils sont représentés par une classe mais leur implantation n'est pas entièrement définie par cette classe.

La hiérarchie des classes représentant les nombres.

```
1 Magnitude
2   Number ()
3       FixedPoint ('numerator' 'denominator' 'scale')
4       Fraction ('numerator' 'denominator')
5       Integer ()
6           LargeInteger ()
7               LargeNegativeInteger ()
8               LargePositiveInteger ()
9       SmallInteger ()
10      LimitedPrecisionReal ()
11          Double ()
12          Float ()
```

### 4.2.3 L'objet nil.

nil (null en Java) est la valeur par défaut du type référence, donc de toute *lvalue* (variable, case de tableau, etc) non explicitement initialisée.

Elle est l'unique instance (*Singleton*) de la classe `UndefinedObject`, qui possède, et sur laquelle il est possible de définir, des méthodes.

Il est possible d'envoyer un message à nil.

L'exception `NullPointerException` n'existe pas!

```
nil isNil "true"

nil class "UndefinedObject"
```

### Application à l'implantation des collections

La classe `UndefinedObject` offre une alternative originale pour l'implantation de certains types récurifs, par exemple `List` ou `Arbre`.

```
1 type List =
2     Empty |
3     Tuple of Object * List
```

dont certaines fonctions se définissent par une répartition sur les différents constructeurs algébriques :

```
1 length :: List -> Int
2 length Empty          = 0
3 length Tuple val suite = 1 + (length suite)
```

### Mise en oeuvre (extrait 1)

```
Object subclass: #List
  instanceVariableNames: 'val suite'
  classVariableNames: ''
  poolDictionaries: ''
  category: 'TP3'!

!List methodsFor: 'accessing' !
suite
    ^suite !

first
    ^val !
```

```

first: element suite: uneListe
    val := element.
    suite := uneListe.!!

!List methodsFor: 'manipulating' !

addFirst: element
    ^List new first: element suite: self !

length
    ^1 + suite length !

append: aList
    ^(self suite append: aList) addFirst: self first !!

```

---

### Mise en oeuvre (extrait 2)

```

!UndefinedObject methodsFor: 'ListManipulation'!
addFirst: element
    ^List with: element !

length
    ^0 !

append: aList
    ^aList !

```

---


#### 4.2.4 Définition de nouvelles structures de contrôle

Les structures de contrôle sont définies en Smalltalk par des méthodes définies sur les classes **Boolean** (conditionnelles, et, ou), **Block** (“tantque”), **Integer** (boucle “for”).

Exercice : ajouter au système les méthodes `ifNotTrue:`, `ifNotFalse:`, `repeatUntil`.

### 4.3 Les Méta-classes en Smalltalk

### 4.3.1 Rappel : la solution Objvlisp



**Figure (5)** – Une classe et une méta-classe sont structurellement et fonctionnellement identiques

- Object est la racine de l'arbre d'héritage,
- Class est instance d'elle-même, sous-classe de Object et racine de l'arbre d'instantiation.

L'intérêt de cette solution est de permettre l'association explicite de n'importe quelle métaclasse à une classe C indépendamment de la relation d'héritage entre C et sa superclasse. Une classe peut ainsi être abstraite (instance de la méta-classe AbstractClass) sans que ses sous-classes le soient.

L'inconvénient de cette solution est qu'elle peut poser des problèmes de compatibilité (voir section 4.3.7). Clos laisse au développeur le soin de résoudre ces problèmes éventuels de compatibilité via la définition de fonctions ad.hoc.

### 4.3.2 Motivations de la solution Smalltalk


La solution Smalltalk a été conçue avec deux objectifs<sup>2</sup> :

- Donner la possibilité de définir des méthodes sur les métaclasses tout en rendant les métaclasses invisibles au développeur ou au chef de projet qui n'a pas envie de les voir. Ceci est fait via l'environnement de programmation.
- Empêcher les problèmes de non compatibilité.


2. Il y a donc deux façons de considérer ce système, la première est de l'utiliser sans chercher à entrer dans le détail de sa réalisation et d'apprécier son efficacité. Le bouton "class" du browser est dédié à la programmation des méthodes de classe et il n'est pas utile de comprendre la mécanique interne pour l'utiliser. La seconde est d'aller y voir de plus près pour le plaisir de comprendre, ce qui prend tout son sens dans le cadre d'un cours sur la réflexivité.



### 4.3.3 Coeur du système de métaclasses Smalltalk



**Figure (6)** – *Classes et Métaclasses : Hiérarchies d'héritage et d'instantiation.* (figure : G.Pavillet)



**Figure (7)** – *Classes et Métaclasses : couches d'instantiation.* (figure : G.Pavillet)

Les points clé :

- Toute classe **C** possède une unique métaclassse référencable via l'expression **C class**.
- Les deux hiérarchies d'héritage entre classes et d'héritage entre métaclasses sont isomorphes.
- **Object class** est la racine de la sous-hiérarchie des métaclasses.
- Le graphe d'héritage étant un arbre avec donc une racine unique (**Object**), le lien entre les hiérarchies de classes et de métaclasses est fait au niveau de la métaclassse **Object class** qui hérite de la classe **Class**.

Ceci se comprend, selon la sémantique usuelle<sup>3</sup> du lien d'héritage, ainsi : une `Object class`, par exemple `Object`), est une sorte de classe.

- Il y a 4 niveaux (conceptuels) d'instantiation. Les objets (couche 1) (par exemple un point), instances des classes (par exemple `Point`, instances des métaclasses (par exemple `Point class`), instances d'une unique méta-méta-classe (`Metaclass`).

Le problème de regression infinie dans les descriptions est réglé par le fait que `metaclass` est instance de `metaclass class`, qui est une métaclasse standard de la couche 3, instance de `Metaclass`. `Metaclass class` a exactement le même format que toutes les autres métaclasses.

#### 4.3.4 Structure des classes et des métaclasses (cf. fig. 8)

---

```
1 ProtoObject #()
2   Object #()
3     Behavior #('superclass' 'methodDict' 'format')
4     ClassDescription #('instanceVariables' 'organization')
5       Class #('subclasses' 'name' 'classPool' 'sharedPools' 'environment' 'category')
6         [ ... all the Metaclasses ... ]
7       Metaclass #('thisClass')
```

---

**Figure (8)** – Les classes “Class” et “Metaclass” (et leurs attributs) définissant les classes et les méta-classes. Elles ont en commun tout ce qui est hérité de “ClassDescription”.

La classe `Class` déclare, en fermant transitivement la relation “est-sous-classe-de”, les attributs : “superclass methodDict format instanceVariables organization subclasses name classPool environment category sharedPools”.

Toute instance de `Class`, directe ou indirecte, pourra posséder une valeur pour chacun d’eux. Par exemple la valeur de l’attribut `instanceVariables` de la classe `Point` (instance de `Point class` qui hérite de `Class`) est ‘x y z’. Les variables de classes sont rangées dans l’attribut `classPool`.


La classe `Metaclass` déclare les attributs : “superclass methodDict format instanceVariables organization this-Class”.

`thisClass` (nom peu évocateur) référence la classe, unique instance (schéma *Singleton*) d’une métaclasse dans un contexte donné. `thisClass` est utilisable dans les méthodes d’instance de la classe `Metaclass` qui n’a aucune sous-classe<sup>4</sup>.

---

3. AKO : a kind of, est une sorte de ... une voiture est une sorte de véhicule.

4. Exercice difficile : créer une sous-classe MC de la classe `Metaclass`, qui définirait un nouveau type généraliste de meta-classes. La difficulté n’est pas dans la création de cette classe mais, comme les métaclasses sont créées automatiquement par le système, de modifier le système de création des classes pour qu’une classe qui en relève ne soit pas une instance de `Metaclass` mais de MC.



**Figure (9)** – l’Inspection d’une classe montre que ses champs sont conformes à la déclaration des attributs réalisés dans la classe `Class`.

#### 4.3.5 Variable de classe et variable d’instance de métaclasse

Un point clé est la différence entre variable de classe et variable d’instance<sup>5</sup> de métaclasse et ce d’autant plus qu’il n’y a pas de différence entre méthode de classe et méthode d’instance de métaclasse.

Les méthodes de classe d’une classe `C` sont les méthodes d’instance de `C class`. Les variables de classes d’une classe `C` n’ont rien à voir avec les variables d’instance de `C class`.

Les variables de classe sont accessibles dans les méthodes d’instance et dans les méthodes de classe. Elle servent à stocker des valeurs communes à toutes les instances d’une même classe, sans avoir à passer par la métaclasse.

Les variables d’instance de métaclasse ne sont accessibles que dans les méthodes de classe. Ce sont les attributs standard des métaclasses considérées comme des classes standard. Si l’on souhaite réaliser une “memo classe” qui mémorise la liste des ses instances, la solution naturelle est d’utiliser une variable d’instance de métaclasse.

#### 4.3.6 Utilisation standard des méta-classes : création d’instances initialisées

Une classe complète avec ses méthodes de création d’instances : la classe `Pile`.

```
1 Object subclass: #Pile
2   instanceVariableNames: 'index buffer '
3   classVariableNames: ''
4   poolDictionaries: ''
5   category: 'Pile'!
6 ...
```

5. On rappelle que le terme “variable d’instance” dénote ce qui est maintenant usuellement appelé “attribut” dans les autres langages.

```

1  "-----"!
2  Pile class
3      instanceVariableNames: ''!
4
5  !Pile class methodsFor: 'creation'!
6  example
7      "self example"
8      "Rend une pile d'entiers de taille 10, pleine."
9      | aStack |
10     aStack := self new: 10.
11     1 to: aStack size do: [:i | aStack push: i].
12     ^aStack!
13
14  new
15      "On fixe la taille par défaut à 10"
16      ^super new initialize: 10!
17
18  new: taille
19      ^super new initialize: taille!
20
21  taille: taille
22      ^self new: taille!!

```

#### 4.3.7 Avantage du modèle Smalltalk : la compatibilité des métaclasse

La solution Smalltalk ne pose pas de problèmes de compatibilité.

Les problèmes de compatibilité surviennent avec les architectures à la `ObjvLisp` lors de phase d'abstraction (passage au niveau meta - compatibilité ascendante) ou de la phase de réification (passage au niveau réifié - compatibilité descendante).

#### Problème de compatibilité Ascendante

Soient :

- une classe A, sa methode `foo` : `'self class bar'`
- une classe B sous-classe de A
- la classe de A et sa méthode `bar`
- la classe de B, non sous-classe de la classe de A, et ne définissant pas `bar`
- une instance b de B

alors : `'b foo'` lève une exception.

Cette situation est impossible avec le modèle Smalltalk où la classe de B ne peut pas ne pas être une sous-classe de la classe de A.

#### Problème de Compatibilité Descendante

Métaclasse MA, méthode `bar` : `'self new foo'`

Métaclasse MB, sous classe de MA,

'mb bar' peut lever une exception.

#### 4.3.8 Inconvénient du modèle Smalltalk

Il est impossible de découpler les hiérarchies de classes et de méta-classes. Ce qui pose des problèmes d'expression. Par exemple si on crée une méta-classe `AbstractClass` dont les instances ne peuvent être instanciées (par une redéfinition appropriée de `new`). Alors toutes les sous-classes d'une instance de `AbstractClass` hériteront de cette redéfinition de `new`. Or, les classes abstraites ont bien sûr des sous-classes concrètes.

### 4.4 Les Méta-objets permettant d'accéder à la pile d'exécution

Smalltalk permet de manipuler chaque bloc (frame) de la pile d'exécution comme un objet de première classe avec une politique de "si-besoin" car la réification est une opération coûteuse.

Par exemple, le code suivant implante les structures de contrôle `catch` et `throw` permettant de réaliser des échappements à la Lisp (réalisation de branchements non locaux), que l'on peut voir comme les structures de contrôle de base nécessaires à l'implantation de mécanismes de gestion des exceptions.

`catch` permet de définir un point de reprise à n'importe quel point d'un programme et `throw` interrompt l'exécution standard et la fait reprendre à l'instruction qui suit le `catch` correspondant.

Application : interruption d'une recherche dès que l'on a trouvé l'élément recherché pour revenir à un point antérieur de l'exécution du programme.

```
1 !Symbol methodsFor: 'catch-throw'!  
  
3 catch: aBlock  
4     "execute aBlock with a throw possibility"  
5     aBlock value!  
  
7 throw: aValue  
8     "Look down the stack for a catch, the mark of which is self,  
9     when found, transfer control (non local branch)."  
10    | catchMethod currentContext |  
11    currentContext := thisContext.  
12    catchMethod := Symbol compiledMethodAt: #catch:..  
13    [currentContext method == catchMethod and: [currentContext receiver == self]]  
14    whileFalse: [currentContext := currentContext sender].  
15    thisContext sender: currentContext sender.  
16    ^aValue! !
```

Exemple d'utilisation, dans un "workspace" :

```
1 #Essai catch: [  
2     Transcript show: 'a';cr.  
3     Transcript show: 'b';cr.  
4     Transcript show: 'c';cr.  
5     #Essai throw: 22.  
6     Transcript show: 'd';cr.  
7     33  
8 ]
```

---

## 4.5 Méta-objets pour accéder au compilateur et aux méthodes compilées

Le compilateur Smalltalk est écrit en Smalltalk et utilisable dans tout programme. Cela permet de définir dynamiquement de nouvelles classes et méthodes.

L'exemple suivant est extrait de la réalisation d'un mini-tableur en Smalltalk. Dans cet exemple, les formules associées aux cellules sont représentées par des méthodes définies dynamiquement par le programme, analysées lexicalement et compilées à la volée

Pour savoir comment ajouter, une fois compilée, une méthode à une classe, il faut étudier les protocoles définis sur les classes `Behavior` et `ClassDescription`.

```
1 Model subclass: #Cellule
2   instanceVariableNames: 'value formula internalFormula dependsFrom '
3   classVariableNames: ''
4   poolDictionaries: ''
5   category: 'Tableur'!

1 !Cellule methodsFor: 'compile formula'!
2 compileFormula: s
3   "Analyse lexicale, puis syntaxique puis generation de code pour la formule s"
4   | tokens newDep interne methodNode |
5   tokens := Scanner new scanTokens: s.
6   newDep := (tokens select: [:i | self isCaseReference: i]) asSet.
7   interne := 'execFormula\ | '.
8   newDep do: [:each | interne := interne , each , ' '].
9   interne := interne , '|\' '.
10  newDep do: [:each | interne := interne , each , ' := (Tableur current at: #' , each , ')
    value.\ '].
11  interne := (interne , '^' , s) withCRs asText.
12  methodNode := UndefinedObject compilerClass new
13    compile: interne
14    in: UndefinedObject
15    notifying: nil
16    ifFail: [].
17  internalFormula := methodNode generate.
18  ^newDep!

1 !Cellule methodsFor: 'exec formula'!
2 executeFormula
3   formula isNil
4   ifFalse:
5     [UndefinedObject addSelector: #execFormula withMethod: internalFormula.
6      ^nil execFormula]
7   ifTrue: [self error]!

9 update: symbol
```

10

```
symbol == #value ifTrue: [self setValue: self executeFormula]!
```